Metal Commands: Section "I"

IF

Type: logical flow command

IF condition_is_true {THEN} commands {:ELSE commands}

IF life=0 THEN goto YourDead

IF bit3(x) and (name$="Wilson") print x*j/byte8(x):ELSE print x*j/byte2(j)

The IF is one of the most commonly used commands in Metal (and in most other languages, for that matter). Using the IF command (and the optional ELSE command) allows you to setup conditional programming.

For example, say that you are going through a for-next loop from 1 to 50, and every time you get to a multiple of 5, you want to print a dot out:

 for j = 1 to 50

  IF j mod 5 = 0 print "."

 next

But what if you want to print a dot out every number but multiplies of 5 - you want to print out asterisks instead? Okay, easy enough:

 for j = 1 to 50

  IF j mod 5 = 0 print "*":ELSE print "."

 next

What we did was modify the IF command line by saying, in effect, the following:

 "If the remainder of J divided by 5 is zero (ie: J is a even multiple of 5), print out an asterisk. However, if the remainder wasn't zero (ie: J is not an even multiple of 5), print out a dot instead."

Using the If-Then is relatively simple. There are just a few rules to remember:

1) Starting with the IF command, and to the end of that line, the rest of the line is considered to be the If-Then line, and will be treated specially.

2) If the condition_is_true expression evaluates to zero (ie: the numerical expression of the condition is mathematically zero), then the command(s) following the optional THEN command will not be run.

3) If the condition_is_true expression evaluates to a non-zero value, then the command(s) following the optional THEN command will be run.

4) If the optional ELSE command is not in the If-Then line, then all happens as expected.

5) If the optional ELSE command is in the If-Then line, then the following two rules apply:

5a) If the value of condition_is_true was zero, then the commands following the ELSE will be run.

5b) If the value of condition_is_true was non-zero, then the commands following the ELSE will not be run.

It does seem rather complex if this is your first foray into programming, but it's not too difficult - Metal is very forgiving using the If-Then commands.

Notice that Metal doesn't care if you use Then in the If-Then or not - Metal is "smart" enough to figure out what you mean either way you go. You can put Then in the program, of course - it's up to you.

Programming notes:

You can nest the If-Then with or without the LongIf-Then up to a maximum of eight (8) times. However, the Gosub command is handled specially in that each Gosub/Return is treated as if they have their own If-Then "nest".

The Goto command is handled as follows:

  If-Then: ends the If-Then

  LongIf-Then: does not end the LongIf-Then

See also: CLEAR, ELSE, END, LONG IF, FOR, GOTO, GOSUB, NEXT, POP, STEP, THEN, UNTIL, WHILE

INPUT

Type: complex command

INPUT {#device{,}} {MODE(mode$)} {[allow$]) {LEN(maxlen)} {"prompttext"} {(prompt$)} {CR} {([echoback$])} variable {TO{,}} {, or CR} {variable(s)}

INPUT i$

INPUT #1,a$

INPUT #2 a$ cr b$ cr d,os$

INPUT mode ("UE ") i$

INPUT len(8) a$

INPUT "Enter your name: " len(32) mode("U") name$

INPUT cr (pr$) zx$

INPUT "Enter password: " len(16) mode("UN ") (["?!@"]) pass$

INPUT ["123456789#-+"] i$

INPUT mode("0A") ["#-+"] i$

INPUT #1,a$(1) to a$(9)

When we say "complex command", we're not kidding - INPUT is one of the most horrifyingly complex commands that Metal has to offer, with up to twelve optional parameters.

As you can see by the examples, INPUT allows you extreme control of what kind of data to input, where to get it from, how much to get, etc. Everything from forcing the input from the keyboard into uppercase to inputting an array from a disk file.

One at a time, here are the optional parameters. Note that for the most part, the actual order of the parameters in the program is not a factor.

Parameter         Does

---------         ----

#device         By using this, you are telling the INPUT command to take input from some place other than the default Device value (normally, this is zero, but you can change this if you wish). You may or may not follow the #device with a comma - it is suggested that you do so, just to make your program code clearer to others.

MODE(mode$)     The MODE optional allows you to change the pre-processing done on typed text from the user - it has no effect on the input from a disk file. The default values are: no case conversion, Escape not allowed, blank lines allowed, leading spaces allowed, commas allowed, all characters allowed.

                  The mode$ is scanned for characters and the following ones that appear in the mode$ set the Mode up. Note that order is not a factor.

                    U : Convert all text from lowercase to uppercase

                    E : Allow Escape to be used, returns chr$(27) in the returned string.

                    N : Do not allow a blank line to be entered.

                 space: Do not allow a leading space to be entered (first character must be a non-space character)

                    , : Do not allow commas to be entered anywhere in the string (disallow comma).

                    A : Only allow A through Z to be entered, complain if anything else is typed.

                    0 : Only allow 0 through 9 to be entered, complain if anything else is typed.

                    / : Only allow "ProDOS Filenames" to be typed - that is, A through Z, 0 through 9, dots (".") and slashes ("/"). Complains if anything else is typed.

[allow$]        Normally, INPUT will allow the user to type in any and all characters. However, by using the optional [allow$] parameter, you can limit the use to what he can type in. Normally, MODE is used to define commonly used sets; however, you can replace the default (or add to the MODE) characters by giving them in a string bracketed with [ ] (square brackets).

[-allow$]       This is similar to [allow$], but the - will remove the characters in the string from the default or custom allow-only-these-keys string. This is called "disallow".

LEN(maxlen)     The LEN optional allows you to specify the maximum number of characters to be typed - normally, this defaults to 255. Once the user has typed the maxlen amount of characters, any further typing will cause the INPUT to complain.

"prompttext"    The text string is simply printed out, with no return printed after it. The semi-colon is implied.

(prompt$)       The string within the parenthesis is printed out, with no return printed after it. The semi-colon is implied. Note that you may use any legal Print expression within the parenthesis.

CR                A CR before the variable prints out a carriage return (chr$13) to both ends.

([echoback$])   Normally, INPUT defaults into echoing back what the user types to the screen. However, there are instances where you don't wish this to happen (like when you are asking for the user's password or something like that). In this case, you can supply a string to be shown one character at a time instead of what the user types. Note that once the user types more letters than there are characters in the echoback$, INPUT will roll-around and re-use the string again. Has no effect during disk-device input.

variable        This is what the whole INPUT mess is about - the variable is not optional, you must have at least one in the INPUT command!

                  If the variable is a string, then INPUT will use whatever it got as input from the device (either the user or a file device) and place it in the string.

                  If the variable is a integer, then INPUT will take the value of the typed/read string, similar to using the Val(string$) command.

,variable2      A comma after the first variable tells the Input command to process the typed line in manner such that it splits up the data so that the first part is before the comma, the second after the comma, etc. (You can have more than two variables in an INPUT line separated by commas)

CR variable2    Using the CR command to separate variables tells the Input command that a hard return (chr$13) follows each data line. Simple way of getting several lines of stuff from a file, or letting the user get a little confused.

TO variable2    The TO command is used to Input an array range. You use it as follows:

                    INPUT array(start) TO array(end)

                  Note that the first and last variables must be the same name - ie: a$(1) and a$(9), name$(8) and name$(30), etc.

                  INPUT will input lines of data from the device starting with the first array and bumping the array number by one until it is finished. For example, if a$(1) TO a$(9) is used, then Input will get nine lines of data and place them in a$(1), a$(2), a$(3)... to a$(9).

Note that INPUT will clear the Abort$ value when it prints out a prompt string and/or takes input from the keyboard/user. It does not destroy the Abort$ value if it is taking input from a file device.

See also: ABORT$, CLEAR, CR, DEVICE, GET, PRINT, SYSINFO

INSTRING

Type: math function using string data

result=INSTRING(findthis$,searchthis${,wildcard$})

a=INSTRING("THE",qe$)

if INSTRING("@",name$) print "You can't use @ in your name!"

x=INSTRING("*SHK",fn$,"*")

INSTRING is one of the commands that return a numerical value based on string data. In the case of INSTRING, the result is a number from 0 to 255.

What INSTRING does is takes a look at the searchthis$ and sees if the string given for findthis$ is somewhere in it. INSTRING doesn't care about the "case" of the strings - it converts both strings, internally, to upper-case before performing the search. After INSTRING finishes looking at the strings, it returns a value to result.

If the value of result is zero, then INSTRING couldn't find the findthis$ inside of searchthis$.

If the value of result is not zero, then INSTRING was able to find the findthis$ inside of searchthis$, and that value returned is the character number into the searchthis$ string.

For example:

print INSTRING("AP","APPLE")       result is 1 - first character of APPLE

print INSTRING("ap","GRAPE")       result is 3 - 3rd character of GRAPE

print INSTRING("AP","Pear")        result is 0 - AP not in PEAR

print INSTRING("ap","tree")        result is 0 - AP not in TREE

As you can see, the upper/lower case of either string is not important.

INSTRING also supports a "wildcard" mode, similar to what the Unix language has for string matching. To use this wildcard mode, you must supply INSTRING with the character it is to use. Normally, you will be using "*" or "=", since they are the ones used in most other systems. However, you can use whatever character you wish.

In the wildcard mode, the result is either 1 (match found) or 0 (no match found).

If the findthis$ is only one character long and is the wildchar$, then the INSTRING will, of course, return a value of 1 - match found.

Examples:

  print INSTRING("*SHK","core.shk","*")      result is 1 - SHK is the last three characters.

  print INSTRING("*SHK","core.bxy","*")      result is 0 - SHK is not the last three characters.

  print INSTRING("*SHK","core.shk1","*")     result is 0 - SHK is not the last three characters (although it does appear in the string).

  print INSTRING("C*","core.shk","*")        result is 1 - C is the first character.

  print INSTRING("C*","scores.shk","*")      result is 0 - C is not the first character (although is does appear in the string).

  print INSTRING("*.*","core.shk","*")       result is 1 - "." does appear in the string.

  print INSTRING("*SHK*","core.shk",*")      result is 1 - SHK does appear in the string.

  print INSTRING("*SHK*","core.shk1","*")    result is 1 - SHK does appear in the string.

See also: ABS, ASC, BITVAL, HEXVAL, LEN, MEMAREA, MEMSIZE, PEEK, PEEKADDR, PEEKBYTE, PEEKWORD, PADDLE, RANDOM, ROTL, ROTR, SIGN, SYSINFO, VAL

